Leveling by Excel

By Steve Y. W. Lam, C.L.S
Department of Land Surveying & Geo-Informatics - Hong Kong Polytechnic University

ABSTRACT

Excel is a powerful yet easy to use spreadsheet program. It can be used to perform survey computations, to generate impressive charts, graphs, and databases, and to share information data with other computer software packages. This paper demonstrates an application of Microsoft® Excel in differential leveling.

LEVELING

Differential leveling, or simply leveling, is usually employed in vertical control surveys. The process is concerned with the determination of height differences between particular points, and indirectly, the altitude of points above or below a reference datum. Instruments which may be used in leveling are digital level, auto-optical level, laser level, vertical graduated staff and palm-top computer.

In leveling, a level is setup approximately midway between point A and point B to determine the height difference between A and B as shown in Figure 1. By reading the graduation on staff A on the horizontal line of sight through the level, reading a is recorded. By reading the graduation on staff B on the horizontal line of sight through the level, reading b is recorded. Height difference between A and B is given by:

In the field, observation data and known bench mark levels are entered into an Excel worksheet.

$$h = a - b \quad \text{where } a = \text{backsight staff reading}, \; b = \text{foresight staff reading}$$

Backsight is the staff reading at the start bench mark or the last turning point after setting up a leveling instrument. Foresight is the staff reading at the closing bench mark or the forward turning point. A bench mark is a permanent point of known or published level.

If we have n setups (Figure 2), i.e. (n – 1) turning points, between A and B, the height difference between A and B is given by:

$$h = (a_1 - b_1) + (a_2 - b_2) + \ldots\ldots + (a_n - b_n)$$
$$= \Sigma_{i=1 \text{ to } n} (a_i - b_i)$$
$$= \Sigma_{i=1 \text{ to } n} (a_i) - \Sigma_{i=1 \text{ to } n} (b_i)$$
$$= \text{sum of backsight readings} - \text{sum of foresight readings}$$

In the field, observation data and known bench mark levels are entered into an Excel worksheet. Format of booking observation data and reduction of the levels are given in Table 1. It is a normal practice to start from a known bench mark and end at another known bench mark to check the quality of the survey. The misclosure is checked against the limits of permissible error of $\pm 12\sqrt{K}$ mm for ordinary leveling, where K is the total distance of a level line in kilometers. If the survey error is acceptable, the misclosure run is then distributed in proportion to the number of instrument setups.



Figure 1 - *Differential leveling between two points at a level setup*



Figure 2 - *Differential leveling between two points with successive setups*

LEVELING PROGRAM BY EXCEL

An Excel file is called a workbook. Each workbook has 16 worksheets. Each worksheet has rows and columns that intersect to form boxes, called cells, into which you can enter your text or numeric data.

Worksheets of Excel use formulas, functions and VBA(Visual BASIC for Applications) macros to perform calculations on the data you have entered. Formulas typically comprise of one or more cell addresses and/ or values and a mathematical operator. For example, if you want cell D1 displays the average value of cells A1, B1 and C1, simply enter the formula =(A1+B1+C1)/3 in the cell as shown in Figure 3).



Figure 3 - *Example of applying formula.*

Worksheets of Excel use formulas, functions and VBA macros to perform calculations on the data you have entered.

Excel includes hundreds of built-in functions for financial, statistical, mathematical, trigonometric, and other types of calculations. Each function has two parts: the name of the function and its arguments. For example, AVERAGE(A1, B1, C1) applies the AVERAGE function to calculate the average values of arguments contained in cells A1, B1 and C1. Arguments may be numbers, cell references, worksheet names, or other data that a function needs to calculate the required value.

Excel macros are written in a programming language called VBA(Visual BASIC for Applications). Using the Visual Basic Editor, which is available in the Tools on the menu bar, you are able to program the procedures and sequence of data processing for more complicated applications.

In Table 1, observation data are entered in the workbook and computed by Excel formulas programmed in the corresponding cells. Formulas of the cells are given in Table 2. Excel does not display the formula in a cell but displays the result of the calculation. You can click on the formula check box in the Tools Menu to view formulas in the cells. In programming a leveling spreadsheet, relative cell references in the formulas are used so that Excel adjusts the cell references in the formulas relative to their new positions in the worksheet. For example, cell G7 contains the formula =IF(E7"","",E7+F7) will automatically change to =IF(E8"","",E8+F8) when G7 is copied by Drag and Drop into G8, thus saving a lot of time typing different formulas.

The survey algorithm given in this example is suitable for many setups in a leveling operation.

Unlike relative cell references, an absolute reference such as $G$5 is a cell reference in a formula that does not change when copied to a new position.

Table 1 - *Format of observation data and leveling result in an Excel worksheet.*

	A	B	C	D	E	F	G	H
1								
2								
3	**Start BM level:**		22.471	(m)	**Total Distance:**		0.25	(km)
4	**End BM level:**		24.246	(m)	**Allowable Error:**		0.006	(m)
5					**Misclosure:**		-0.005	(m)
6					**Number of setups:**		9	
7					**[BS] – [FS]:**		1.770	
8								
9	**BS**	**IS**	**FS**	**Height of Collimation**	**Reduced Level**	**Adjust -ment**	**Final Level**	**Remarks**
10	0.618			23.089	22.471		22.471	Geodetic BM152
11	1.465		0.658	23.896	22.431	0.001	22.432	TP
12	1.088		0.411	24.573	23.485	0.001	23.486	TP
13	0.226		0.366	24.433	24.207	0.002	24.209	TP
14	0.044		1.732	22.745	22.701	0.002	22.703	TP
15	0.862		1.864	21.743	20.881	0.003	20.884	Iron Bar
16		1.275			20.468	0.003	20.471	Manhole cover
17		1.823			19.920	0.003	19.923	Kerb top
18		1.888			19.855	0.003	19.858	Kerb bottom
19	1.712		0.620	22.835	21.123	0.003	21.126	TP
20	1.560		0.263	24.132	22.572	0.004	22.576	TP
21	0.822		0.077	24.877	24.055	0.004	24.059	TP
22			0.636		24.241	0.005	24.246	Geodetic BM126

Table 2 - *Formulas for reduction of levels in Table 1*

	D	E
	Height of Collimation	**Reduced Level**
10	=IF(A10="","",A10+E10)	=C3
11	=IF(A11="","",A11+E11)	=IF(B11&C11="","",IF(B11="",E10+A10-C11,IF(C11="",E10+A10-B11)))
12	=IF(A12="","",A12+E12)	=IF(B12&C12="","",IF(B12="",E11+A11+B11-C12,IF(C12="",E11+A11+B11-B12)))
13	=IF(A13="","",A13+E13)	=IF(B13&C13="","",IF(B13="",E12+A12+B12-C13,IF(C13="",E12+A12+B12-B13)))
14	=IF(A14="","",A14+E14)	=IF(B14&C14="","",IF(B14="",E13+A13+B13-C14,IF(C14="",E13+A13+B13-B14)))
15	=IF(A15="","",A15+E15)	=IF(B15&C15="","",IF(B15="",E14+A14+B14-C15,IF(C15="",E14+A14+B14-B15)))
16	=IF(A16="","",A16+E16)	=IF(B16&C16="","",IF(B16="",E15+A15+B15-C16,IF(C16="",E15+A15+B15-B16)))
17	=IF(A17="","",A17+E17)	=IF(B17&C17="","",IF(B17="",E16+A16+B16-C17,IF(C17="",E16+A16+B16-B17)))
18	=IF(A18="","",A18+E18)	=IF(B18&C18="","",IF(B18="",E17+A17+B17-C18,IF(C18="",E17+A17+B17-B18)))
19	=IF(A19="","",A19+E19)	=IF(B19&C19="","",IF(B19="",E18+A18+B18-C19,IF(C19="",E18+A18+B18-B19)))
20	=IF(A20="","",A20+E20)	=IF(B20&C20="","",IF(B20="",E19+A19+B19-C20,IF(C20="",E19+A19+B19-B20)))
21	=IF(A21="","",A21+E21)	=IF(B21&C21="","",IF(B21="",E20+A20+B20-C21,IF(C21="",E20+A20+B20-B21)))
22	=IF(A22="","",A22+E22)	=IF(B22&C22="","",IF(B22="",E21+A21+B21-C22,IF(C22="",E21+A21+B21-B22)))
23	=IF(A23="","",A23+E23)	=IF(B23&C23="","",IF(B23="",E22+A22+B22-C23,IF(C23="",E22+A22+B22-B23)))
24	=IF(A24="","",A24+E24)	=IF(B24&C24="","",IF(B24="",E23+A23+B23-C24,IF(C24="",E23+A23+B23-B24)))

	F
	Adjustment
10	
11	=IF(B11&C11="","",IF(C11="",-$G$5/$G$6*(COUNT($C$10:C11)+1),-$G$5/$G$6*COUNT($C$10:C11)))
12	=IF(B12&C12="","",IF(C12="",-$G$5/$G$6*(COUNT($C$10:C12)+1),-$G$5/$G$6*COUNT($C$10:C12)))
13	=IF(B13&C13="","",IF(C13="",-$G$5/$G$6*(COUNT($C$10:C13)+1),-$G$5/$G$6*COUNT($C$10:C13)))
14	=IF(B14&C14="","",IF(C14="",-$G$5/$G$6*(COUNT($C$10:C14)+1),-$G$5/$G$6*COUNT($C$10:C14)))
15	=IF(B15&C15="","",IF(C15="",-$G$5/$G$6*(COUNT($C$10:C15)+1),-$G$5/$G$6*COUNT($C$10:C15)))
16	=IF(B16&C16="","",IF(C16="",-$G$5/$G$6*(COUNT($C$10:C16)+1),-$G$5/$G$6*COUNT($C$10:C16)))
17	=IF(B17&C17="","",IF(C17="",-$G$5/$G$6*(COUNT($C$10:C17)+1),-$G$5/$G$6*COUNT($C$10:C17)))
18	=IF(B18&C18="","",IF(C18="",-$G$5/$G$6*(COUNT($C$10:C18)+1),-$G$5/$G$6*COUNT($C$10:C18)))
19	=IF(B19&C19="","",IF(C19="",-$G$5/$G$6*(COUNT($C$10:C19)+1),-$G$5/$G$6*COUNT($C$10:C19)))
20	=IF(B20&C20="","",IF(C20="",-$G$5/$G$6*(COUNT($C$10:C20)+1),-$G$5/$G$6*COUNT($C$10:C20)))
21	=IF(B21&C21="","",IF(C21="",-$G$5/$G$6*(COUNT($C$10:C21)+1),-$G$5/$G$6*COUNT($C$10:C21)))
22	=IF(B22&C22="","",IF(C22="",-$G$5/$G$6*(COUNT($C$10:C22)+1),-$G$5/$G$6*COUNT($C$10:C22)))
23	=IF(B23&C23="","",IF(C23="",-$G$5/$G$6*(COUNT($C$10:C23)+1),-$G$5/$G$6*COUNT($C$10:C23)))
24	=IF(B24&C24="","",IF(C24="",-$G$5/$G$6*(COUNT($C$10:C24)+1),-$G$5/$G$6*COUNT($C$10:C24)))

	G
4	=12*SQRT(G3)/1000
5	=C3+SUM(A10:A23)-SUM(C11:C24)-C4
6	=COUNT(A10:A23)
7	=SUM(A10:A23)-SUM(C11:C24)
8	
9	**Final Level**
10	=C3
11	=IF(E11="","",E11+F11)
12	=IF(E12="","",E12+F12)
13	=IF(E13="","",E13+F13)
14	=IF(E14="","",E14+F14)
15	=IF(E15="","",E15+F15)
16	=IF(E16="","",E16+F16)
17	=IF(E17="","",E17+F17)
18	=IF(E18="","",E18+F18)
19	=IF(E19="","",E19+F19)
20	=IF(E20="","",E20+F20)
21	=IF(E21="","",E21+F21)
22	=IF(E22="","",E22+F22)
23	=IF(E23="","",E23+F23)
24	=IF(E24="","",E24+F24)

When new rows are inserted for more setups, make sure to copy the formulas of the previous row and update the range of rows for the formulas given in cells G5, G6 and G7.

The survey algorithm given in this example is suitable for many setups in a leveling operation. However, the current maximum number of rows in an Excel worksheet is 16384 and you can enter as many as 255 characters per cell. For more applications and programming techniques, readers could refer Excel's operating manuals, programming manuals and reference materials.

Author Steve Y. W. Lam, CLS, can be reached at:
Department of Land Surveying and Geo-informatics
Hong Kong Polytechnic University
Hung Hom, Kowloon, Hong Kong
Tel.: (852) 2766 5964
Fax: (852) 2330 2994
E-mail: 1slams@polyu.edu.hk
Web: http:// www.1sgi.polyu.edu.hk/